THE DELTA-T TUNING PROCEDURE FOR THE FERMILAB LINAC UPGRADE

Thomas L. Owens
Fermi National Accelerator Laboratory
P.O. Box 500
Batavia, Illinois 60510

Abstract

The analysis necessary to perform the delta-t procedure on the upgraded linac at Fermilab is described. Out of the analysis has come three simple subroutines with which module phase and amplitude can be calculated. Subroutine inputs include experimental measurements of delta-t values, variation of energy with module phase, and peak energy change through a module. Two separate delta-t methods are required to tune all of the modules in the linac upgrade. The accuracy and stability of each method has been defined as a function of linac module.

I. Introduction

The delta-t procedure was developed at Los Alamos National Laboratory many years ago for the purpose of tuning the phase and amplitude of each of 44 accelerator modules on the LAMPF linear accelerator.[1] Tests of the delta-t procedure on the existing 200 MeV linac at Fermilab show experimental trends that can be predicted by theory.[2] The theory of the delta-t procedure for the 200 MeV linac at Fermilab was described in an earlier report.[3] The theory and computer codes developed in reference 3 have been modified recently to apply to the upgraded linac. The modifications will be described in this report.

The principal difference in the linacs, which necessitates the modifications in computer codes, is that each module of the existing linac consists of a single Alvarez tank, while the upgraded modules are side coupled structures divided into four sections. Each section is separated by a drift space. The Alvarez tanks have cell lengths equal to $\beta\lambda$ while each section of the side-coupled structures has cell lengths equal to $\langle\beta\rangle\lambda/2$, where $\langle\beta\rangle$ is the beta in the center of each of the four sections of a module. Some of the points in the analysis where these differences are important will be highlighted in this report. Throughout the analysis, the upgraded linac modules will be identified by numbers 11-17 to distinguish them from the old linac tanks, which are numbered 1-5.

II. Analysis for the Linac Upgrade

The transformation of phase and beta across each cell of the linac are given by the following expressions:

$$\phi_e = \phi_i + l_1 \phi_i' + l_2 (\phi_i' + \Delta\phi) + \delta\phi \tag{1}$$

$$\Delta W = e V \cos (\phi_o) \left[T_o - 2\pi (\frac{\beta_g}{\beta_c} - 1) T_p \right] \tag{2}$$

$$\phi_o = \phi_i + l_1 \phi_i' + \delta\psi \tag{3}$$

$$\beta_c = \beta_i + \frac{e V}{2 \beta_i \gamma_i^3 W_o} \left[T_o \cos (\phi_o) + S_o \sin (\phi_o) \right] \tag{4}$$

where,

$$\phi_i' = \frac{d\phi_i}{dz} = \frac{2\pi}{\beta_i \lambda}$$

$$\Delta\phi = - \frac{2\pi \Delta W}{\lambda \beta_i^3 \gamma_i^3 W_o}$$

$$\delta\phi = - \frac{2\pi e V \beta_g}{\beta_o^3 \gamma_o^3 W_o} T_p \sin(\phi_o)$$

$$d\psi = - \frac{\pi e V \beta_g}{\beta_h^3 \gamma_h^3 W_o} \left[T_p \sin(\phi_o) + S_p \cos(\phi_o) \right]$$

$$\beta_o = \frac{\beta_i + \beta_e}{2} \quad , \quad \beta_h = \frac{\beta_i + \beta_c}{2} \quad , \quad \beta_g = \frac{2L}{\lambda}$$

The transit time factors are defined as follows,

$$T_o = \frac{1}{V} \int_{-\frac{L}{2}}^{\frac{L}{2}} E(z) \cos(kz)\, dz$$

$$S_o = \frac{2}{V} \int_{0}^{\frac{L}{2}} E(z) \sin(kz)\, dz$$

$$T_p = \frac{k}{2\pi V} \int_{-\frac{L}{2}}^{\frac{L}{2}} z\,E(z)\,\sin(kz)\,dz$$

$$S_p = \frac{k}{\pi V} \int_{0}^{\frac{L}{2}} z\,E(z)\,\cos(kz)\,dz$$

where $k = 2\pi / \beta_g\lambda$, $\phi_{e,i}$ are the exit and entrance phases for the cell, ΔW is the energy change through the cell, ϕ_o is the phase at the center of the accelerating gap, l_1 is the distance from the cell entrance to the center of the gap, l_2 is the distance from the center of the gap to the exit of the cell, V is the voltage across the gap, W_o is the rest energy of the particle, and L is the cell length. The center of the accelerating gap is at z=0. Comparing the above equations for a $\beta\lambda/2$ structure to the equations for a $\beta\lambda$ structure (reference 3), the principal differences in the equations lie in the definitions and use of the quantities, β_g , k, and V.

As mentioned in the introduction, each module in the linac upgrade consists of four sections separated by a drift space. The cell length of each cell within a section is constant and equal to $<\beta>\lambda/2$. The drift space between each section is $3\beta\lambda/2$ in length. The overall transformation of input phase and energy is calculated by repeated application of the above equations in this geometry. The original computer subroutine for calculating the transformation in an Alvarez tank was called XFER in reference 3. This subroutine has been modified to account for the electrical and geometric differences between an Alvarez tank and the side coupled structure used in the linac upgrade. The new subroutine is called XFERS. The general convention that will be followed when a code from reference 3 has been modified will be to add the letter, s, to the end of the original program or subroutine name. A listing of the subroutine XFERS is given in Appendix A.

The analysis of the delta-t procedure begins with the identification of a design particle based upon the longitudinal dynamics contained in equations 1-4. The same dynamics will be used in all aspects of the delta-t procedure to achieve

the best accuracy. In the case of the new linac, the design particle is intended to have a phase of approximately -32 degrees, relative to the peak field, in the center of the first and last cells of each section. The program XYZS uses the transformation subroutine, XFERS, to identify the input phase and energy which minimizes the RMS variation of phase in the first and last cells of each section of each module of the linac. The IMSL subroutine DBCONF has been used to perform the numerical minimization.

The file, SYNCS.DAT contains a listing of the phases and betas at the module inputs and outputs which have minimized the RMS phase variations in the manner just described. Phases are given in radians. The module input phases listed are those found at the input to the first cell of each module, rather than values at the centers of the cells. Phases at the centers of the first cell of each module are listed in the last column of the file, SYNCS.DAT. The central phases for the design particle are found to be slightly less than -32 degrees for each module. The design betas agree to better than 4 decimal places with values generated during the original design of the upgraded linac.[4]

As described in reference 3, once the design particle has been identified in the program XYZS, the transfer matrix, **M**, can be calculated. This matrix and elements of the matrices **T**, **A**, and **B** are calculated in the program DTPARS (see reference 1 or 3 for definitions of these matrices). A listing of the elements of these matrices is given in Table I in the files MS.DAT, TMATS.DAT, AMATS.DAT, and BMATS.DAT. Module and beam monitor distances, needed to calculate matrix elements, are contained in the input file DS.DAT. Beam monitor distances, given in the file, have been supplied by the engineering department and are preliminary, at present. No engineering plans currently exist for monitor placement after the last module of the linac. In the preliminary positions for these monitors shown in file DS.DAT, no interference occurs with any other object along the beam pipe, according to information supplied by Carol Johnstone. Carol Johnstone is managing the construction of the transfer line from the end of the linac to the booster.

The changes in the times of flight for the design particles when modules are alternately turned on then off are given in the file TABS.DAT, also calculated in the program DTPARS. The corresponding phases at a frequency of 805 MHz are given in TABS.DAT. Since the phase detectors used in the procedure operate at 201 MHz, the phase values listed in file, TABS.DAT, should be divided by four when the procedure is applied to the upgraded linac. The phase values in file, TABS.DAT, are used to set the zero in the delta-t plane for the upgraded linac modules. Accuracy of the delta-t procedure depends critically upon the accuracy of this zero setting in the delta-t plane.

Once the matrices **M**, **T**, **A**, and **B** have been calculated, errors and stability of the delta-t methods can be calculated in a manner similar to the analysis for the

existing linac described in reference 3. Errors in output energy are calculated assuming a 13.8 picosecond error in the time measurement (one degree phase error). The error calculation is performed in the program DTERRS. The stability calculation is performed in the program DTSTABS. Both programs are listed in the appendix. Results from code runs are shown in figures 1 and 2. The figures demonstrate that method 1 would work best in modules 11 and 12, while method 2 would work best in modules 13-17. Experimental procedures for each method are outlined in the next two sections.

III. Method #1 Experimental Procedures

As described in reference 3, the displacement from design of the electric field magnitude is first determined. In modules 11 and 12, where method 1 is recommended, either of two techniques can be used to determine electric field magnitude. Applying the first technique, the slope of a line generated in the delta-t plane as phase is varied is measured near the design phase. The electric field is related to the slope of this line. The design phase, about which the slope is measured, is normally within a few degrees of the point where curves of various electric field values intersect. The file, SLOPES.DAT, lists the values of the design slopes, S, for all of the modules in the linac upgrade. These values have been calculated using program, SLOPES, listed in the appendix.

Applying a second technique to determine electric field magnitude, the energy change through the module is measured as the phase is varied. The electric field can be related to the slope of this curve near the design phase. Slope values for the design particle are also provided in the program, SLOPES. Slope values are listed in the file SLOPES.DAT under the column heading, M(2,1), since this slope is just the 2,1 element of the matrix, **M**. The derivatives of the slopes with respect to electric field magnitude are also given in the file, SLOPES.DAT, for both field estimating techniques.

If the slope of the curve of energy change versus phase is measured in the experiment, the measured slope can be input to subroutine, EFSETS. This subroutine provides an estimate of the electric field displacement from design. Corrections to the electric field should be made at this point.

The initial delta-t values are next measured. Any inaccuracy in the location of the zero in the delta-t plane will affect the accuracy of this measurement. The measured values of Δt_B and Δt_C are input to the subroutine, PHSETS. This subroutine outputs an estimate of the energy and phase displacements from

design. Corrections to the module phase can then be made. The process can be repeated to improve accuracy.

IV. Method #2 Experimental Procedures

Method 2 is recommended for tuning linac upgrade modules 13-17. In this case, the electric field displacement is estimated by measuring the peak energy change through a module as the module phase is changed. A technique for making the energy change measurement using the beam monitors was described in reference 3. The electric field is next changed one or two percent and the new value of peak energy change is recorded. The two values of peak energy change and the fractional change in electric field is input to the subroutine, EFSET2S. With these experimental inputs, the subroutine provides an estimate of the electric field displacement.

Values for the peak energy change of the design particle, input to the subroutine, EFSET2S, are calculated within the program, WPEAKS. A program listing of WPEAKS is given in the appendix. The program also outputs to a plot file values of energy change as a function of module phase. A sample plot is given in figure 3 for module 11. A distinct energy peak is obtained for all modules of the linac upgrade. The file WPEAKS.LIS, shown in Table I, lists the values of the peak energy displacement and the phase displacement where the peak energy occurs for modules 11-17.

A target line is next generated in the delta-t plane and displayed in some fashion. The target line is given by the equation,

$$a_{22}\,\Delta t_B - a_{21}\,\Delta t_C = 0$$

where the a's are elements of the matrix, **A**, described in section II and in reference 3. Values for the matrix elements are given in Table I in the file, AMATS.DAT, for the linac upgrade modules. The phase of each module is next varied until the line generated in the delta-t plane intersects the target line. The correct phase setting occurs at this intersection, as described in references 1 and 3. To improve accuracy, the electric field measurement is repeated and additional adjustments of the module phase are made.

V. Summary

The delta-t procedure can be performed on the upgraded linac using only three simple subroutines, EFSETS, EFSET2S, and PHSETS and their associated data files. The data files contain the basic parameters needed to tune the linac. These files have been generated from extensive calculations of longitudinal dynamics for the linac upgrade. Using the data files and a small number of experimental inputs, the subroutines, EFSETS and EFSET2S, are used to adjust the electric field magnitudes in the modules. The subroutine, PHSETS, is used to adjust the phase. The data file AMATS.DAT contains parameters which define a target line which must be intersected by varying the phase of the module. The procedures outlined in this report should be reasonably straight-forward and accurate, in principle. Considerable care will have to be taken in hardware development to insure accuracy.

References

1. K. R. Crandall, "The Delta-T Tuneup Procedure for the LAMPF 805 MHz Linac," LANL Report LA-6374-MS, June, 1976.

2. T. L. Owens and E. S. McCrory, "The Delta-T Tuneup Procedure For the Fermilab Linac," in *Proceedings of the 1990 Linear Accelerator Conference*, Albuquerque, New Mexico, September 10-14, 1990 (Also Fermilab Report CONF-90/207).

3. T. L. Owens, "Phase and Amplitude Tuning Procedures for the Fermilab Linac," Fermilab Report TM-1713, January 14, 1991.

4. J. A. MacLachlan, "Transition Section Design Rationale and New Parameters," Fermilab Linac Upgrade Document LU-158, April 30, 1990.

Table I. Data files generated or used by the delta-t programs.

```
$ TYPE SYNCS.DAT
MODULE    PHASE IN      PHASE OUT      BETA IN       BETA OUT      PHASE CNTR

 11     -.212972D+01  -.212588D+01  0.456922D+00  0.509412D+00  -.536133D+00
 12     -.213197D+01  -.212741D+01  0.509397D+00  0.555343D+00  -.543210D+00
 13     -.213136D+01  -.212937D+01  0.555344D+00  0.595604D+00  -.546123D+00
 14     -.212920D+01  -.212684D+01  0.595601D+00  0.630912D+00  -.546561D+00
 15     -.212642D+01  -.212535D+01  0.630904D+00  0.662051D+00  -.545818D+00
 16     -.213120D+01  -.212922D+01  0.662048D+00  0.689443D+00  -.552182D+00
 17     -.213000D+01  -.212832D+01  0.689456D+00  0.713778D+00  -.552192D+00
$
$
$ TYPE TABS.DAT
MODULE      TAB            TAC          PHASE AB      PHASE AC

 11     0.268475D-08  0.834991D-08  0.778008D+03  0.241970D+04
 12     0.211037D-08  0.651648D-08  0.611559D+03  0.188840D+04
 13     0.169369D-08  0.521586D-08  0.490810D+03  0.151149D+04
 14     0.138444D-08  0.425293D-08  0.401195D+03  0.123245D+04
 15     0.115393D-08  0.353221D-08  0.334395D+03  0.102359D+04
 16     0.966198D-09  0.295318D-08  0.279993D+03  0.855795D+03
 17     0.823582D-09  0.245982D-08  0.238664D+03  0.712825D+03
$
```

```
$ TYPE MS.DAT
MODULE      M11             M21             M12             M22

 11     -.855916D+00   -.171645D+01   0.455496D-01   -.107754D+01
 12     -.907894D+00   -.521155D+00   -.687625D-02   -.110573D+01
 13     -.894681D+00   0.122804D+01   -.401122D-01   -.106287D+01
 14     -.826311D+00   0.331866D+01   -.601692D-01   -.968696D+00
 15     -.727413D+00   0.548493D+01   -.703652D-01   -.844272D+00
 16     -.620146D+00   0.757753D+01   -.732747D-01   -.717271D+00
 17     -.500210D+00   0.963737D+01   -.736366D-01   -.580502D+00
$
$
$ TYPE TMATS.DAT
MODULE      T11             T21             T12             T22

 11     0.365148D-09   0.143851D-09   -.183950D-09   -.520201D-09
 12     0.376809D-09   0.326262D-09   -.124633D-09   -.371122D-09
 13     0.375273D-09   0.468086D-09   -.867260D-10   -.270514D-09
 14     0.362502D-09   0.563076D-09   -.617370D-10   -.200013D-09
 15     0.343572D-09   0.613753D-09   -.449405D-10   -.149679D-09
 16     0.322669D-09   0.632408D-09   -.334596D-10   -.113913D-09
 17     0.299041D-09   0.619951D-09   -.251811D-10   -.853818D-10
$
$
$ TYPE AMATS.DAT
MODULE      A11             A21             A12             A22

 11     0.318187D+10   0.879878D+09   -.112515D+10   -.223347D+10
 12     0.374193D+10   0.328961D+10   -.125664D+10   -.379927D+10
 13     0.444038D+10   0.768345D+10   -.142357D+10   -.615995D+10
 14     0.529944D+10   0.149190D+11   -.163575D+10   -.960466D+10
 15     0.627765D+10   0.257413D+11   -.188484D+10   -.144097D+11
 16     0.730392D+10   0.405489D+11   -.214537D+10   -.206890D+11
 17     0.860562D+10   0.624847D+11   -.253800D+10   -.301403D+11
$
$
$ TYPE BMATS.DAT
MODULE      B11             B21             B12             B22

 11     -.268333D+10   -.640962D+10   0.861302D+09   0.433791D+10
 12     -.341989D+10   -.558753D+10   0.116702D+10   0.485585D+10
 13     -.428093D+10   -.271356D+10   0.152073D+10   0.479905D+10
 14     -.527665D+10   0.313505D+10   0.192955D+10   0.387549D+10
 15     -.637773D+10   0.126998D+11   0.238500D+10   0.182748D+10
 16     -.750071D+10   0.262611D+11   0.284642D+10   -.141700D+10
 17     -.890578D+10   0.466630D+11   0.348896D+10   -.696309D+10
$
```

```
$ TYPE SLOPES.DAT
TANK     SLOPE=S          dS/dE          M(2,1)       dM(2,1)/dE

 11     0.393953D+00  -.223049D+01  -.171645D+01  -.602367D+01
 12     0.865855D+00  -.160808D+01  -.521155D+00  -.623720D+01
 13     0.124732D+01  -.121529D+01  0.122804D+01  -.591253D+01
 14     0.155330D+01  -.951005D+00  0.331866D+01  -.510134D+01
 15     0.178639D+01  -.768181D+00  0.548493D+01  -.400763D+01
 16     0.195993D+01  -.625183D+00  0.757753D+01  -.276639D+01
 17     0.207313D+01  -.507298D+00  0.963737D+01  -.141312D+01
$
$
$ TYPE WPEAKS.LIS
MODULE /WPEAK (MEV) /DEL PHI (DEG)

   11  0.257380D+01  0.460700D+02
   12  0.330811D+01  0.471480D+02
   13  0.399045D+01  0.469140D+02
   14  0.459384D+01  0.457730D+02
   15  0.513143D+01  0.443670D+02
   16  0.572610D+01  0.436170D+02
   17  0.615315D+01  0.422580D+02
$
```

```
        D1              D2          DAB
$ TYPE DS.DAT
  11 6.113780D-02 7.592395D+00 6.572800D+00
  12 6.634480D-02 8.199231D+00 7.241490D+00
  13 6.200140D-02 8.737128D+00 7.827000D+00
  14 6.433820D-02 9.215649D+00 8.340070D+00
  15 7.200900D-02 9.633321D+00 8.791190D+00
  16 7.498080D-02 1.000012D+01 9.188510D+00
  17 7.500000D-02 1.000000D+01 9.539960D+00 ! D1,D2 AS OF 3/20/91.
$
$                       E
$ TYPE GS.DAT
  11 8.049664D+08 8.070000D+00 1.000000D-07
  12 8.049664D+08 7.850000D+00 1.000000D-07
  13 8.049664D+08 7.660000D+00 1.000000D-07
  14 8.049664D+08 7.480000D+00 1.000000D-07
  15 8.049664D+08 7.340000D+00 1.000000D-07
  16 8.049664D+08 7.200000D+00 1.000000D-07
  17 8.049664D+08 7.090000D+00 1.000000D-07

$
$
$ TYPE SS.DAT
  11 1.600000D+01 -1.220240D02 0.456922D+00
  12 1.600000D+01 -1.221530D02 0.509397D+00
  13 1.600000D+01 -1.221180D02 0.555344D+00
  14 1.600000D+01 -1.219940D02 0.595601D+00
  15 1.600000D+01 -1.218350D02 0.630904D+00
  16 1.600000D+01 -1.221090D02 0.662048D+00
  17 1.600000D+01 -1.220400D02 0.689456D+00

$
```


OUTPUT ENERGY UNCERTAINTY IN UPGRADE FOR 13.8 PS RANDOM ERROR
(CIRCLES - METHOD #1 UNOPTIMIZED TARGET, TRIANGLES - METHOD #2 OPTIMIZED TARGET)
ENERGY UNCERTAINTY (PERCENT)
0.30
0.25
0.20
0.15
0.10
0.05
0.00
11
12
13
14
15
16
17
LINAC UPGRADE MODULE NUMBER

Figure 1


STABILITY RATIO FOR METHOD #2 IN THE FERMILAB UPGRADE
2.5
2.0
1.5
1.0
0.5
0.0
STABILITY RATIO
11
12
13
14
15
16
17
LINAC UPGRADE MODULE NUMBER

Figure 2


ENERGY DISPLACEMENT VERSUS MODULE PHASE
TI (MODULE 11, E=8.07 MEV/M, PEAK ENERGY=2.5738 MEV, PHASE AT PEAK=46.070 DEGREES)
ENERGY DISPLACEMENT (MEV)
3.0
2.5
2.0
1.5
1.0
0.5
0.0
-0.5
0
10
20
30
40
50
MODULE PHASE DISPLACEMENT (DEGREES)

Figure 3

Appendix A

Source Files and Subroutines Used in the Delta-T Analysis

```
      PROGRAM XYZS
C
C*****************************************************************
C
C  CALCULATES THE BEST VALUES FOR SYNCHRONOUS BETA AND PHASE
C  INPUT TO THE FIRST CELL OF EACH OF THE ACCELERATOR TANKS.
C  ASSUMES CELL LENGTH=BETA*LAMDA/2, AND ONE MODULE IS COMPOSED OF
C  4 SECTIONS, EACH HAVING CONSTANT CELL LENGTHS.  THE NUMBER OF
C  CELLS PER SECTION IS GIVEN BY PARAMETER, FNCEL.
C
C  VALUES ARE CALCULATED FROM THE GEOMETRY OF THE TANK
C  AND ARE INDEPENDENT OF THE PARTICLE DYNAMICS WHICH GENERATED
C  THE GEOMETRY, ORIGINALLY.
C
C  THE BEST BETA AND PHASE ARE THOSE WHICH MINIMIZE THE
C  RMS VARIATION OF THE PHASE AT THE CENTERS OF THE CELLS
C  IN THE FIRST AND LAST CELLS OF EACH SECTION OF A MODULE.
C
C  THE FOLLOWING DATA FILES ARE REQUIRED:
C
C       GENLS.DAT       GENERAL INPUT DATA(TANK NUMBER,FREQUENCY,ETC.)
C
C       TXX.DAT         TANK "XX" GEOMETRY (XX="11"-"17" FOR THE LINAC
C                       UPGRADE MODULES, CONTAINS DRIFT LNGTHS & CELL LNGTHS).
C
C
C       CF.DAT          TANK "XX" COEFFICIENTS FOR TRANSIT TIME FACTOR
C                       FITS.
C
C  REQURES SUBROUTINES:
C       FUNXY
C       XFERS
C
C  BEST BETA AND PHASE INPUT TO MODULE IS WRITTEN TO FILE "XYZS.OUT"
C
C  USES IMSL ROUTINE "DBCONF" TO FIND THE MINIMUM OF THE RMS VARIATION OF
C  THE PHASE IN THE FIRST AND LAST CELLS OF THE SECTIONS, AS A FUNCTION
C  OF INPUT BETA AND PHASE IN THE MODULE.
C
C  MUST BE COMPILED AND LINKED ON THE FNAL NETWORK, SINCE PROGRAM
C  ACCESSES THE IMSL LIBRARY.  ONCE COMPILED AND LINKED, XYZS.EXE
C  CAN BE RUN ON THE ADCALC NETWORK.
C
C  * THIS ROUTINE IS A MODIFICATION OF CODE XYZ
C
C                   *** WRITTEN BY T.L. OWENS ***
C                         AUGUST 8, 1990
C
C                      MODIFIED MARCH 18, 1991
C
C*****************************************************************
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      EXTERNAL FUNXY,DBCONF
      PARAMETER (NC=100)
      DIMENSION IPARAM(7),RPARAM(7),XS(2),XGUESS(2),XSCALE(2)
      DIMENSION XLB(2),XUB(2)
      DIMENSION PCEL(NC),BCEL(NC),PCNTR(NC),SEP(4),CLN(4)
      CHARACTER TANK*2,TFILE*7
```

```
      COMMON/PARAM/FREQ,EREST,PI,TPI,C,WAVL,TW,FNCEL
      COMMON/GEND/EF0
      COMMON/TNO/TANK
      DATA XSCALE/1.0D0,1.0D0/,FSCALE/1.0/
      OPEN(UNIT=11,FILE='GENLS.DAT',TYPE='OLD')
      READ(11,2)TANK,FREQ,ESYNC,PSYNC,BSYNC,FNCEL,XLB,XUB
2     FORMAT(10X,A2,/,9(10X,D13.6,/))
      CLOSE(UNIT=11)
      NCEL=FNCEL
      C=2.99792458D8
      PI=3.14159265358979
      TPI=2.0*PI
      WAVL=C/FREQ
      TW=TPI/WAVL
C
C***  NEGATIVE ION REST ENERGY.
C***  PROTON EREST WOULD BE 938.2796.
C
      EREST=939.25
C
C***  CALCULATE INITIAL VALUE FOR SYNC BETA AND PHASE AT TANK INPUT.
C***  REQUIRES CELL LENGTHS TO ESTIMATE BINI SO MUST READ TXX.DAT FILES.
C
      TFILE='T'//TANK//'.DAT'
      OPEN(UNIT=11,FILE=TFILE,STATUS='OLD')
      READ(11,227)(SEP(J),CLN(J),J=1,4)
227   FORMAT(2(2X,F6.5))
      CLOSE(UNIT=11)
      BINI=2.0*CLN(1)/WAVL
      PINI=PSYNC*PI/180.0
      EF0=ESYNC
C
C*** CALCULATE THE SYNCHRONOUS BETA AND PHASE BY MINIMIZING
C*** RMS PHASE (RELATIVE TO PHASE IN CELL ONE) AT THE FIRST AND LAST
C*** CELLS OF EACH SECTION OF A MODULE.
C
      XGUESS(1)=BINI
      XGUESS(2)=PINI
      N=2
C
C*** CALL MINIMIZATION ROUTINE (IMSL LIBRARY)
C
      CALL DBCONF(FUNXY,N,XGUESS,0,XLB,XUB,XSCALE
     1,FSCALE,IPARAM,RPARAM,XS,FV)
C
C*** WRITE BEST VALUES FOR BETA AND PHASE INPUT TO FILE "XYZS.OUT".
C
      OPEN(UNIT=11,FILE='XYZS.OUT',STATUS='NEW')
C     WRITE(5,15)BSYNC,PSYN*180.0/PI
      WRITE(11,15)XS(1),XS(2)*180.0/PI,FV,(IPARAM(L),L=3,5)
15    FORMAT(' BSYNC=',D12.6,' PSYN=',D12.6,' FUNXY=',D12.6
     1,//,' NO. ITERS=',I3,' NO. EVALS=',I3,' NO GRAD EVALS=',I3)
      STOP
      END
C
```

```
      SUBROUTINE FUNXY (N,XS,FV)
C
C*** FUNXY IS THE RMS DEVIATION FROM THE CENTRAL PHASE IN THE FIRST CELL
C*** OF THE CENTRAL PHASES IN THE FIRST AND LAST CELLS OF EACH SECTION
C*** OF A MODULE.
C
C     INPUTS ARE:
C         XS(1) =         BETA INTO TANK
C         XS(2) =         PHASE INTO TANK
C
C                  ***  MODIFIED BY T.L. OWENS  ***
C                         MARCH 11, 1990
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      PARAMETER (NC=100)
      CHARACTER TANK*2
      DIMENSION BCEL(NC),PCEL(NC),PCNTR(NC),XS(N)
      COMMON/GEND/EF0
      COMMON/TNO/TANK
      COMMON/PARAM/FREQ,EREST,PI,TPI,C,WAVL,TW,FNCEL
      NCEL=FNCEL
      CALL XFERS(TANK,XS(1),XS(2),EF0,BCEL,PCEL,PCNTR)
      PRMSS=0.0
      DO 1,I=1,4
      PRMSS=PRMSS+(PCNTR((I-1)*NCEL+1)-PCNTR(1))**2
1     PRMSS=PRMSS+(PCNTR(I*NCEL)-PCNTR(1))**2
      FV=PRMSS
      RETURN
      END
C
```

```
      SUBROUTINE XFERS(TANK,BSTRT,PSTRT,EF0,BCEL,PCEL,PCNTR)
C
C
C*********************************************************************
C
C  SUBROUTINE XFERS CALCULATES BETA AND PHASE AT THE INPUTS TO EACH CELL
C  OF A LINAC MODULE (4 SECTIONS, JOINED BY 3 DRIFT SPACES).
C  A (BETA-LAMDA)/2 STRUCTURE IS ASSUMED.  THE PHASE AT THE CENTER OF EACH GAP
C  IS ALSO CALCULATED.  CELL LENGTHS FOR EACH CELL OF A SECTION ARE ASSUMED
C  CONSTANT.  NUMBER OF CELLS PER SECTION IS GIVEN BY PARAMETER FNCEL.
C
C  INPUTS ARE:
C       BSTRT   BETA INTO FIRST CELL.
C       PSTRT   PHASE INTO FIRST CELL (RADIANS). NOTE THAT THIS PHASE
C               SHOULD BE ABOUT -122 DEGREES FOR FERMI BETA LAMDA/2 STRUCTURE.
C       EF0     MODULE ELECTRIC FIELD (MV/M).
C       TANK    MODULE IDENTIFIER ("11"-"17" FOR LINAC UPGRADE MODULES)
C               MUST BE DECLARED CHAR TANK*2 IN CALLING PROGRAM.
C
C  OUTPUTS ARE:
C       BCEL    ARRAY CONTAINING BETA VALUES AT EACH CELL INPUT.
C       PCEL    ARRAY CONTAINING PHASE VALUES AT EACH CELL INPUT.
C       PCNTR   ARRAY CONTAINING PHASE AT CENTER OF EACH CELL.
C
C       NOTE: IF THERE ARE N CELLS, THE OUTPUT BETA AND PHASE OF THE
C               MODULE WILL BE CONTAINED IN BCEL(N+1) & PCEL(N+1).
C
C               IF THERE ARE M CELLS PER SECTION, THE OUTPUT BETA OF THE
C               SECTION WILL BE CONTAINED IN BCEL(M+1).  PCEL(M+1)
C               CONTAINS ONLY THE PHASE INPUT TO THE NEXT SECTION.
C
C   * THIS SUBROUTINE IS A MODIFICATION OF THE EARLIER XFER SUBROUTINE
C     WHICH WAS WRITTEN FOR ALVAREZ STRUCTURES HAVING BETA*LAMDA=CELL LNGTH.
C
C
C                     ***  WRITTEN BY T. L. OWENS  ***
C                             MARCH 5, 1991
C
C*********************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      PARAMETER (NC=100)
      CHARACTER TANK*2,TFILE*7
      COMMON/PARAM/FREQ,EREST,PI,TPI,C,WAVL,TW,FNCEL
      COMMON/GLN/CLN(4)
      DIMENSION BCEL(NC),PCEL(NC),PCNTR(NC),T(4),S(4),TP(4),SP(4)
     1,TC(4),SC(4),TPC(4),SPC(4),SEP(4),BETAG(4)
      TFILE='T'//TANK//'.DAT'
      OPEN(UNIT=11,FILE=TFILE,STATUS='OLD')
      READ(11,1)(SEP(J),CLN(J),J=1,4)
1     FORMAT(2(2X,F6.5))
      CLOSE(UNIT=11)
      OPEN(UNIT=11,FILE='CF.DAT',STATUS='OLD')
      READ(11,2)(TC(J),SC(J),TPC(J),SPC(J),J=1,4)
2     FORMAT(4(2X,F8.5))
      CLOSE(UNIT=11)
      BINI=BSTRT
      PINI=PSTRT
```

```
        NCELLS=FNCEL
        CPR=TW*EF0/EREST
        DWSUM=0.0
        DBETASUM=0.0
        GAMAI=DSQRT(1.0/(1.0-BINI**2))
        WINI=EREST*(GAMAI-1.0)
        PI0=PINI+PI/2.0
        BCEN=0.0
        BAV=BINI
        GAV=DSQRT(1.0/(1.0-BAV**2))
        BCEL(1)=BINI
        PCEL(1)=PINI
C
C*** CALCULATE TRANSIT TIME FACTORS AND GEOMETRIC BETAS.
C
        DO 4 I=1,4
        BTG=2.0*CLN(I)/WAVL
        BETAG(I)=BTG
        BTS=BTG**2
        BTC=BTG*BTS
        T(I)=TC(1)+TC(2)*BTG+TC(3)*BTS+TC(4)*BTC
        S(I)=SC(1)+SC(2)*BTG+SC(3)*BTS+SC(4)*BTC
        TP(I)=TPC(1)+TPC(2)*BTG+TPC(3)*BTS+TPC(4)*BTC
        SP(I)=SPC(1)+SPC(2)*BTG+SPC(3)*BTS+SPC(4)*BTC
4       CONTINUE
C
C*** MAIN LOOP THROUGH MODULE. SECTION INDEX IS J. CELL INDEX IS I.
C
        DO 6 J=1,4
        DO 5 I=1,NCELLS
        GINI=DSQRT(1.0/(1.0-BINI**2))
        WINI=EREST*(GINI-1.0)
        FL1=CLN(J)/2.0
        FL2=FL1
        P1=TW/BINI*FL1
        VG=EF0*CLN(J)
C
C*** LOOP TO DETERMINE CENTRAL BETA
C
        JFLAG=0
7       FPR=PI*VG*BETAG(J)/((BAV*GAV)**3*EREST)
C
C*** LOOP TO SOLVE TRANSCENDENTAL EQUATION FOR CENTRAL PHASE
C
        IFLG3=0
33      PI0S=PI0
        T1=FPR*(TP(J)*DSIN(PI0S)+SP(J)*DCOS(PI0S))
        PI0=PINI+P1-T1
        IFLG3=IFLG3+1
        IF(IFLG3.EQ.1)GO TO 33
        IF(ABS(1.0-PI0/PI0S).LT.1.0E-10)GO TO 30
        IF(IFLG3.EQ.20)GO TO 32
        GO TO 33
32      WRITE(5,34)
34      FORMAT(' ETRAN> CENTRAL PHASE DOES NOT CONVERGE-EXITING')
        GO TO 1000
30      CONTINUE
C
```

```
C*** END LOOP
C
        BCENS=BCEN
        BCEN=BINI+VG/(2.0*BINI*GINI**3*EREST)
     1*(T(J)*DCOS(PI0)+S(J)*DSIN(PI0))
        BAV=(BINI+BCEN)/2.0
        GAV=DSQRT(1.0/(1.0-BAV**2))
        JFLAG=JFLAG+1
        IF(JFLAG.EQ.1)GO TO 7
        IF(ABS(1.0-BCEN/BCENS).LT.1.0E-10)GO TO 8
        IF(JFLAG.GT.20)GO TO 9
        GO TO 7
9       WRITE(5,100)
100     FORMAT(' CENTRAL BETA VALUE DOES NOT CONVERGE-EXITING')
        GO TO 1000
C
C*** END LOOP
C
8       CONTINUE
        PCNTR(I+(J-1)*NCELLS)=PI0
        DWTRM=TPI*(BETAG(J)/BCEN-1.0)*TP(J)
        DW=VG*DCOS(PI0)*(T(J)-DWTRM)
        DWSUM=DWSUM+DW
        BT1=1.0+(WINI+DW)/EREST
        BINIS=BINI
        BINI=DSQRT(1.0-1.0/BT1**2)
        BBAV=(BINIS+BINI)/2.0
        GGAV=DSQRT(1.0/(1.0-BBAV**2))
        F2=TW/((BINI*GINI)**3*EREST)
        T2=FL2*(P1/FL1-F2*DW)
        T3=TPI*VG*BETAG(J)/((BBAV*GGAV)**3*EREST)*TP(J)*DSIN(PI0)
        PHIE=PINI+P1+T2-T3
156     PINI=PHIE-PI
        BCEL(I+1+(J-1)*NCELLS)=BINI
        PCEL(I+1+(J-1)*NCELLS)=PINI
5       CONTINUE
        IF(J.EQ.4)GO TO 1000
        PINI=PINI+TPI*FREQ*SEP(J+1)/(BINI*C)-3.0*PI
        PCEL(1+J*NCELLS)=PINI
6       BCEL(1+J*NCELLS)=BINI
C
C*** END OF MAIN LOOP
C
1000    RETURN
        END
$
```

```
      PROGRAM DTPARS
C
C*****************************************************************
C
C  PROGRAM TO GENERATE TABLES OF THE TRANSFER MATRICES,M , AND THE VARIOUS
C  TRANSIT MATRICES, T, A, AND B FOR ALL LINAC TANKS.  THESE MATRICES
C  ARE USED IN SUBSEQUENT CALCULATIONS OF DELTA-T PROPERTIES.
C
C  INPUT DATA FILES REQUIRED ARE:
C
C     GS.DAT            GENERAL TANK DATA-TANK,FREQ,ESYNC,EPS
C     DS.DAT            BPM DISTANCES DAB,D1,D2
C     SS.DAT            SYNCRONOUS VALUES-NO. CELLS,PHASE IN ,BETA IN
C
C  OUTPUT DATA FILES ARE:
C
C     TABS.DAT          TAB,TAC,AND CORRESPONDING PHASES
C     SYNCS.DAT         SYNCHRONOUS VALUES-BETA IN,OUT;PHASE IN,OUT
C     MS.DAT            TRANSFER MATRIX,M, ELEMENTS
C     TMATS.DAT         TRANSIT MATRIX,T, ELEMENTS
C     AMATS.DAT         INVERSE OT T MATRIX
C     BMATS.DAT         B MATRIX=MA
C
C
C              ***  WRITTEN BY T.L. OWENS  ***
C                     AUGUST 21,1990
C
C      MODIFIED FOR BETA*LAMDA/2 STRUCTURE MARCH 18,1991
C
C*****************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      PARAMETER (NC=100)
      DIMENSION FM(2,2),TMT(2,2),AMT(2,2),BMT(2,2)
     1,BCEL(NC),PCEL(NC),PCNTR(NC)
      CHARACTER TANK*2
      COMMON/PARAM/FREQ,EREST,PI,TPI,C,WAVL,TW,FNCEL
      OPEN(UNIT=3,FILE='DS.DAT',STATUS='OLD')
      OPEN(UNIT=4,FILE='SS.DAT',STATUS='OLD')
      OPEN(UNIT=7,FILE='TABS.DAT',STATUS='NEW')
      OPEN(UNIT=10,FILE='SYNCS.DAT',STATUS='NEW')
      OPEN(UNIT=8,FILE='MS.DAT',STATUS='NEW')
      OPEN(UNIT=20,FILE='TMATS.DAT',STATUS='NEW')
      OPEN(UNIT=21,FILE='AMATS.DAT',STATUS='NEW')
      OPEN(UNIT=22,FILE='BMATS.DAT',STATUS='NEW')
      OPEN(UNIT=23,FILE='GS.DAT',STATUS='OLD')
      WRITE(7,77)
77    FORMAT(' MODULE',5X,'TAB',11X,'TAC',9X,'PHASE AB',6X,'PHASE AC',/)
      WRITE(10,100)
100   FORMAT(' MODULE',4X,'PHASE IN',5X,'PHASE OUT',6X
     1,'BETA IN',5X,' BETA OUT',/)
      WRITE(8,80)
80    FORMAT(' MODULE',5X,'M11',11X,'M21',11X,'M12',11X,'M22',/)
      WRITE(20,200)
200   FORMAT(' MODULE',5X,'T11',11X,'T21',11X,'T12',11X,'T22',/)
      WRITE(21,210)
210   FORMAT(' MODULE',5X,'A11',11X,'A21',11X,'A12',11X,'A22',/)
      WRITE(22,220)
```

```
220     FORMAT(' MODULE',5X,'B11',11X,'B21',11X,'B12',11X,'B22',/)
C
C*** BEGINNING OF LOOP THROUGH TANKS
C
7       READ(3,1,END=1000)TANK,D1,D2,DAB
1       FORMAT(2X,A2,3(1X,D12.6))
        READ(23,40)ITANK,FREQ,ESYNC,EPS
        READ(4,40)ITANK,FNCEL,PHIA,BTA
40      FORMAT(2X,I2,3(1X,D12.6))
C       WRITE(5,40)ITANK,FREQ,ESYNC,EPS
        LCEL=FNCEL*4.0+1.0
        C=2.99792458D8
        PI=3.14159265358979
        PHIA=PHIA*PI/180.0
        TPI=2.0*PI
        OMEGA=TPI*FREQ
        WAVL=C/FREQ
        TW=TPI/WAVL
        EREST=939.25
C
C*** CALCULATE OUTPUT BETA AND PHASE FOR SYNCHRONOUS PARTICLE
C
        CALL XFERS(TANK,BTA,PHIA,ESYNC,BCEL,PCEL,PCNTR)
        BTB=BCEL(LCEL)
        PHIB=PCEL(LCEL)
C
C*** CALCULATE PHASE SETTINGS FOR ZERO IN DELTA-T PLANE
C
        VA=BTA*C
        VB=BTB*C
        TABOF=(DAB+D1)/VA
C
C*** 9.0 BELOW ARISES BECAUSE 3 CELL LENGTHS EXIST BETWEEN EACH SECTION,
C*** AND THERE ARE 3 SUCH DRIFT SPACES PER MODULE.
C
        TABON=(4.0*FNCEL+9.0)/(2.0*FREQ)+D1/VB+(PHIB-PHIA)/(FREQ*TPI)
        DTABD=TABOF-TABON
        TACOF=(DAB+D2)/VA
        TACON=TABON+(D2-D1)/VB
        DTACD=TACOF-TACON
        DPABZ=DTABD*FREQ*360.0
        DPACZ=DTACD*FREQ*360.0
        WRITE(7,6)ITANK,DTABD,DTACD,DPABZ,DPACZ
6       FORMAT(2X,I2,2X,4(2X,D12.6))
        WRITE(10,6)ITANK,PHIA,PHIB,BTA,BTB
C
C* BEGIN CALCULATION OF TRANSFER MATRIX
C
        DELBA=BTA*EPS
        DELPA=PHIA*EPS
        BAPLS=BTA+DELBA
        BAMNS=BTA-DELBA
        GAPLS=DSQRT(1.0/(1.0-BAPLS**2))
        GAMNS=DSQRT(1.0/(1.0-BAMNS**2))
        WAPLS=EREST*(GAPLS-1.0)
        WAMNS=EREST*(GAMNS-1.0)
        DELWA=(WAPLS-WAMNS)/2.0
        PAPLS=PHIA+DELPA
```

```
        PAMNS=PHIA-DELPA
C
C*** TAKE VARIOUS DERIVATIVES W.R.T INPUT PHASE AND ENERGY
C*** TO CALCULATE ELEMENTS OF THE TRANSFER MATRIX
C
        CALL XFERS(TANK,BAPLS,PHIA,ESYNC,BCEL,PCEL,PCNTR)
        BBPWA=BCEL(LCEL)
        PBPWA=PCEL(LCEL)
        CALL XFERS(TANK,BAMNS,PHIA,ESYNC,BCEL,PCEL,PCNTR)
        BBMWA=BCEL(LCEL)
        PBMWA=PCEL(LCEL)
        CALL XFERS(TANK,BTA,PAPLS,ESYNC,BCEL,PCEL,PCNTR)
        BBPPA=BCEL(LCEL)
        PBPPA=PCEL(LCEL)
        CALL XFERS(TANK,BTA,PAMNS,ESYNC,BCEL,PCEL,PCNTR)
        BBMPA=BCEL(LCEL)
        PBMPA=PCEL(LCEL)
        GBPWA=DSQRT(1.0/(1.0-BBPWA**2))
        GBMWA=DSQRT(1.0/(1.0-BBMWA**2))
        GBPPA=DSQRT(1.0/(1.0-BBPPA**2))
        GBMPA=DSQRT(1.0/(1.0-BBMPA**2))
        WBPWA=EREST*(GBPWA-1.0)
        WBMWA=EREST*(GBMWA-1.0)
        WBPPA=EREST*(GBPPA-1.0)
        WBMPA=EREST*(GBMPA-1.0)
C
        FM(1,1)=(PBPPA-PBMPA)/(2.0*DELPA)
        FM(1,2)=(PBPWA-PBMWA)/(2.0*DELWA)
        FM(2,1)=(WBPPA-WBMPA)/(2.0*DELPA)
        FM(2,2)=(WBPWA-WBMWA)/(2.0*DELWA)
C
        WRITE(8,6)ITANK,FM
C
C*** CALCULATE THE T,A, AND B MATRICES
C
        GAMA=DSQRT(1.0/(1.0-BTA**2))
        GAMB=DSQRT(1.0/(1.0-BTB**2))
        WA=EREST*(GAMA-1.0)
        WB=EREST*(GAMB-1.0)
        EX=3.0/2.0
        ETAC=(GAMA**2-1.0)**EX
        ETBC=(GAMB**2-1.0)**EX
        DF=EREST*C
        DFA=DF*ETAC
        DFB=DF*ETBC
        TM1=(1.0-FM(1,1))/OMEGA
        TM2=FM(1,2)/OMEGA
        TM3=FM(2,1)/DFB
        TM4=(1.0/ETAC-FM(2,2)/ETBC)/DF
        TM5=DAB/DFA
C
        TMT(1,1)=TM1+D1*TM3
        TMT(1,2)=-TM2-TM5-D1*TM4
        TMT(2,1)=TM1+D2*TM3
        TMT(2,2)=-TM2-TM5-D2*TM4
C
C*** A MATRIX
C
```

```
      DET=TMT(2,2)*TMT(1,1)-TMT(1,2)*TMT(2,1)
      AMT(1,1)=TMT(2,2)/DET
      AMT(1,2)=-TMT(1,2)/DET
      AMT(2,1)=-TMT(2,1)/DET
      AMT(2,2)=TMT(1,1)/DET
C
C*** B MATRIX
C
      BMT(1,1)=FM(1,1)*AMT(1,1)+FM(1,2)*AMT(2,1)
      BMT(1,2)=FM(1,1)*AMT(1,2)+FM(1,2)*AMT(2,2)
      BMT(2,1)=FM(2,1)*AMT(1,1)+FM(2,2)*AMT(2,1)
      BMT(2,2)=FM(2,1)*AMT(1,2)+FM(2,2)*AMT(2,2)
C
      WRITE(20,6)ITANK,TMT
      WRITE(21,6)ITANK,AMT
      WRITE(22,6)ITANK,BMT
C
      GO TO 7
1000  STOP
      END
C
```

```
      PROGRAM DTERRS
C
C*****************************************************************
C
C  PROGRAM TO CALCULATE THE STABILITY RATIO AND THE UNCERTAINTY
C  IN THE OUTPUT ENERGY FOR THE DELTA-T MEASUREMENTS.
C
C  REQUIRES THE VARIOUS FORMS OF THE TRANSIT TIME MATRIX
C  (T, A, B) AND THE TRANSFER MATRIX, FM.
C
C  THE STABILITY RATIO IS OUTPUT TO FILE "STAB.DAT"
C  AND THE UNCERTAINTY PER 13.8 PS TIME ERROR IS OUTPUT TO
C  FILE "METH1.DAT" FOR DELTA-T METHOD #1 (LOW ENERGY MODULES)
C  AND TO "METH2.DAT" FOR METHOD #2 (HIGH ENERGY MODULES).
C
C             WRITTEN BY T.L. OWENS
C                AUGUST 20,1990
C
C MODIFIED DEC. 12,1990
C
C MODIFIED FOR BETA*LAMDA/2 STRUCTURE MARCH 18, 1991.
C
C*****************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION FM(2,2),T(2,2),A(2,2),B(2,2)
      OPEN(UNIT=11,FILE='MS.DAT',STATUS='OLD')
      OPEN(UNIT=3,FILE='TMATS.DAT',STATUS='OLD')
      OPEN(UNIT=4,FILE='AMATS.DAT',STATUS='OLD')
      OPEN(UNIT=7,FILE='BMATS.DAT',STATUS='OLD')
      OPEN(UNIT=8,FILE='SYNCS.DAT',STATUS='OLD')
      OPEN(UNIT=20,FILE='STABS.DAT',STATUS='NEW')
      OPEN(UNIT=21,FILE='METH2S.DAT',STATUS='NEW')
      OPEN(UNIT=22,FILE='METH1S.DAT',STATUS='NEW')
      READ(3,30)
      READ(4,30)
      READ(7,30)
      READ(8,30)
      READ(11,30)
30    FORMAT(/)
7     READ(11,1,END=1000)ITANK,FM
1     FORMAT(2X,I2,2X,4(2X,D12.6))
      READ(3,1)ITANK,T
      READ(4,1)ITANK,A
      READ(7,1)ITANK,B
      READ(8,1)ITANK,PHIA,PHIB,BTA,BTB
      WRITE(5,1)ITANK,T
      WRITE(5,1)ITANK,A
      WRITE(5,1)ITANK,B
      WRITE(5,1)ITANK,PHIA,PHIB,BTA,BTB
C
C*** SET RANDOM ERROR IN TIME MEASUREMENT TO 13.8 PICOSECONDS.
C
      TERR=1.38D-11
      EREST=939.25
      GAMA=DSQRT(1.0/(1.0-BTA**2))
      GAMB=DSQRT(1.0/(1.0-BTB**2))
      WA=EREST*(GAMA-1.0)
```

```
        WB=EREST*(GAMB-1.0)
        FNUM=T(2,2)/T(1,1)+T(1,2)/T(2,1)
        DENOM=T(2,1)/T(1,1)+T(1,1)/T(2,1)
C
C*** CALCULATE STABILITY RATIO
C
        STABR=(FM(2,2)-FM(2,1)*FNUM/DENOM)*WA/WB
        STABR=DABS(STABR)
C
C*** CALCULATE FRACTIONAL UNCERTAINTY IN OUTPUT ENERGY PER
C*** PICOSECOND ERROR IN MEASUREMENT
C
        S=A(2,1)/A(2,2)
C
C*** DWBS1 IS UNCERTAINTY SQUARED FOR METHOD #1 (LOW ENERGY MODULES)
C*** DWBS IS UNCERTAINTY SQUARED FOR METHOD #2 (HIGH ENERGY MODULES)
C
        S1=-A(1,1)/A(1,2)
        S2=-S
        DWBS1=(B(2,1)+S2*B(2,2))**2*(1.0+S1**2)/(S1-S2)**2*TERR**2
        DWBD1=SQRT(DWBS1)/WB
        DWBS=(B(2,1)-S*B(2,2))**2*(1.0/(S**2+1.0))*TERR**2
        DWBD=SQRT(DWBS)/WB
        FTANK=ITANK
        WRITE(20,2)ITANK,STABR
2       FORMAT(2X,I2,2X,E12.6)
        WRITE(21,2)ITANK,DWBD*100.0
        WRITE(22,2)ITANK,DWBD1*100.0
        GO TO 7
1000    STOP
        END

$
```

```
      PROGRAM WPEAKS
C
C*********************************************************************
C
C  PROGRAM TO GENERATE CURVES OF ENERGY CHANGE VERSUS PHASE
C  DISPLACEMENT FROM SYNCHRONOUS VALUES AT THE OUTPUT OF EACH MODULE
C  OF A BETA-LAMDA/2 STRUCTURE COMPOSED OF 4 SECTIONS.
C
C  THE MAXIMUM ENERGY DISPLACEMENT AND THE CORRESPONDING PHASE
C  DISPLACEMENT ARE FOUND.
C
C  INPUT DATA FILES:
C
C       WPEAKS.DAT
C       SYNCRS.DAT      RANDOM ACCESS FILE CREATED FROM FILE "SYNC.DAT"
C                       CONTAINING SYNCHRONOUS PHASES AND ENERGIES
C                       AT THE INPUT AND OUTPUT OF TANKS.
C
C  OUTPUT FILES:
C
C       WPEAKS.OUT      CONTAINS PLOT VALUES, ENERGY DISPLACEMENT VERSUS
C                       PHASE (FROM SYNCHRONOUS VALUES).
C
C       WDIFS.OUT       CONTAINS PLOT VALUES FOR ENERGY CHANGE DEVIATION
C                       FROM SYNCHRONOUS VALUES. THIS IS THE
C                       QUANTITY THAT WOULD BE MEASURED IN PRACTICE
C
C       WPEAKS.MAX      CONTAINS THE PEAK ENERGY CHANGE AND PHASE AT PEAK
C                       (FOR WPEAK.OUT FILE ONLY CURRENTLY).
C
C  USES SUBROUTINE XFERS.
C
C  NOTE:  OUTPUT IN WPEAKS.OUT IS DISPLACEMENT FROM SYNCHRONOUS ENERGY OUT
C         NOT THE DEVIATION IN THE OUTPUT/INPUT DIFFERENCE FROM SYNC VALUES.
C
C
C                ***   WRITTEN BY T. L. OWENS   ***
C                         OCT. 17,1990
C
C        MODIFIED FOR BETA-LAMDA/2 STRUCTURE MARCH 20, 1991
C
C*********************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION DWB(200),PHI(200),BCEL(100),PCEL(100),PCNTR(100)
     1,DWABS(200)
      CHARACTER TANK*2
      COMMON/PARAM/FREQ,EREST,PI,TPI,C,WAVL,TW,FNCEL
      OPEN(UNIT=8,FILE='WPEAKS.OUT',STATUS='NEW')
      OPEN(UNIT=3,FILE='WPEAKS.DAT',STATUS='OLD')
      OPEN(UNIT=7,FILE='WPEAKS.MAX',STATUS='NEW')
      OPEN(UNIT=20,FILE='WDIFS.OUT',STATUS='NEW')
C
C*** OPEN RANDOM ACCESS FILE TO OBTAIN SYNCHRONOUS PARAMETERS
C
      OPEN(UNIT=4,FILE='SYNCRS.DAT',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      READ(3,11)ITANK
```

```
11      FORMAT(10X,I2)
        READ(3,12)FREQ,EF0,FNCEL,PINT,FNPTS,TOLPH,DPINI,DBINI
12      FORMAT(10X,D12.6)
C
C*** CONVERT INTEGER TANK NUMBER TO CHARACTER VARIABLE
C*** ASCII FOR INTEGERS 0-9 IS 48-57.
C
        IC10=ITANK/10
        IC1=ITANK-IC10*10
        TANK=CHAR(IC10+48)//CHAR(IC1+48)
        READ(4,6,REC=ITANK-10)ITNK,PHIA,PHIB,BTA,BTB
6       FORMAT(2X,I2,2X,4(2X,D12.6))
C
C*** EREST USED BY LANL (SWAIN).
C
        EREST=939.301
        PI=4.0*DATAN(1.0D0)
        TPI=2.0*PI
        RD=180.0/PI
        DR=PI/180.0
        PINT=PINT*DR
        TOLPH=TOLPH*DR
        C=2.99792458D8
        WAVL=C/FREQ
        TW=TPI/WAVL
        LCEL=FNCEL*4.0+1.0
        PHINC=PINT/FNPTS
        NPTS=FNPTS
        PSTRT=PHIA+DPINI*DR
        BSTRT=BTA*(1.0+DBINI)
        IFLAG=0
        GSOUT=DSQRT(1.0/(1.0-BTB**2))
        GSIN=DSQRT(1.0/(1.0-BTA**2))
        GIN=DSQRT(1.0/(1.0-BSTRT**2))
        DWINI=EREST*(GIN-GSIN)
C
C*** OBTAIN POINTS FOR DWB VERSUS PHI PLOTS
C
        DO 15 I=1,NPTS+1
        CALL XFERS(TANK,BSTRT,PSTRT,EF0,BCEL,PCEL,PCNTR)
        GOUT=DSQRT(1.0/(1.0-BCEL(LCEL)**2))
C
C*** DWB IS ENERGY DIFFERENCE BETWEEN OUTPUT AND SYNCHRONOUS OUTPUT ENERGY.
C
        DWB(I)=EREST*(GOUT-GSOUT)
C
C*** DWABS IS DEVIATION FROM SYNC ENERGY CHANGE OF THE TOTAL ENERGY
C*** CHANGE THROUGH THE MODULE.
C*** THIS IS THE QUANTITY THAT WOULD BE MEASURED IN PRACTICE.
C
        DWABS(I)=DWB(I)-DWINI
        PHI(I)=PSTRT
        WRITE(8,51)(PHI(I)-PHIA)*RD,DWB(I)
51      FORMAT(2X,E12.6,2X,E12.6)
        WRITE(20,51)(PHI(I)-PHIA)*RD,DWABS(I)
        PSTRT=PSTRT+PHINC
        IF(I.LE.2.OR.IFLAG.EQ.1)GO TO 15
        IF(DWB(I).LT.DWB(I-1).AND.DWB(I-1).GT.DWB(I-2))THEN
```

```
          P1=PHI(I-2)
          P3=PHI(I)
          D1=DWB(I-2)
          D3=DWB(I)
          IFLAG=1
        END IF
15      CONTINUE
        IF(IFLAG.EQ.0)THEN
          WRITE(5,52)
52        FORMAT(' WPEAK DOES NOT LIE IN PHASE INTERVAL SELECTED
     1 ---- INCREASE SEARCH INTERVAL.')
          GO TO 1000
        END IF
C
C*** FIND PEAK IN ENERGY CURVE.
C
        IFLG=1
        ITMAX=50
C
C*** IF TOLPH IS 0 THEN SKIP THE MAX FINDER
C
        IF(TOLPH.EQ.0.0)GO TO 1000
        DPHI=(P3-P1)/2.0
        DWST=D1
        PSTRT=P1
30      PSTRT=PSTRT+DPHI
        IF(DABS(DPHI).LT.TOLPH)GO TO 900
32      CALL XFERS(TANK,BSTRT,PSTRT,EF0,BCEL,PCEL,PCNTR)
        GOUT=DSQRT(1.0/(1.0-BCEL(LCEL)**2))
        DWBB=EREST*(GOUT-GSOUT)
        IF(DWBB.LT.DWST)THEN
          DPHI=-DPHI/2.0
        END IF
        DWST=DWBB
        IFLG=IFLG+1
        IF(IFLG.GT.ITMAX)THEN
          WRITE(5,33)
33        FORMAT(' WPEAK> MAX FINDER DOES NOT CONVERGE')
          GO TO 1000
        END IF
        GO TO 30
C900    WRITE(7,37)(PSTRT-PHIA)*RD,DWBB
900     WRITE(7,37)(PSTRT-PHIA)*RD,DWBB-DWINI
37      FORMAT(2X,E12.6,2X,E12.6)
C
C*** WRITE PHASE AND ENERGY CHANGE THROUGH TANK AT MAX ENERGY.
C
C       WRITE(5,37)(PSTRT-PHIA)*RD,DWBB-DWINI
1000    STOP
        END
C
```

```
      PROGRAM SLOPES
C
C*************************************************************
C
C  PROGRAM CALCULATES THE SLOPE OF LINE IN DELTA-T PLANE
C  AND THE DERIVATIVE OF THE SLOPE WRT E. ALSO CALCULATES
C  DERIVATIVE OF 2,1 ELEMENT OF THE M MATRIX.
C
C  INPUTS:
C       TMATM.DAT       TMATS VALUES AT E-.
C       TMATP.DAT       TMATS VALUES AT E+.
C       TMATS.DAT       TMATS VALUES AT ESYNC.
C       MM.DAT          MS VALUES AT E-
C       MP.DAT          MS VALUES AT E+
C       MS.DAT          MS VALUES AT ESYNC.
C       GS.DAT          CONTAINS NEEDED VALUES OF ESYNC.
C
C  OUTPUTS:
C       SLOPES.DAT      SLOPES AND DERIVATIVES.
C
C  CURRENTLY E+ = E(DESIGN)*1.005 AND E- = E(DESIGN)*0.995
C
C  THE FILES TMATP.DAT, TMATM.DAT, MP.DAT, MM.DAT ARE CURRENTLY
C  GENERATED BY RUNNING XYZS.EXE WITH E+ AND E-, THEN RUNNING DTPAR.EXE
C  WITH E+ AND E- USED IN THE FILE GS.DAT WHICH INPUTS TO DTPARS.EXE.
C  THE FILES ARE COPIED FROM TMATS.DAT AND MS.DAT AS THEY ARE GENERATED BY
C  DTPARS.EXE.
C
C                    ***   WRITTEN BY T. L. OWENS   ***
C                             MARCH 20,1991
C
C*************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION TMT(4),TMTM(4),TMTP(4),FM(4),FMM(4),FMP(4)
      OPEN(UNIT=4,FILE='TMATM.DAT',STATUS='OLD')
      OPEN(UNIT=7,FILE='TMATP.DAT',STATUS='OLD')
      OPEN(UNIT=3,FILE='MM.DAT',STATUS='OLD')
      OPEN(UNIT=11,FILE='MP.DAT',STATUS='OLD')
      OPEN(UNIT=8,FILE='SLOPES.DAT',STATUS='NEW')
      OPEN(UNIT=12,FILE='TMATS.DAT',STATUS='OLD')
      OPEN(UNIT=14,FILE='MS.DAT',STATUS='OLD')
      OPEN(UNIT=15,FILE='GS.DAT',STATUS='OLD')
      WRITE(8,5)
5     FORMAT(' TANK','    SLOPE=S','        dS/dE','         M(2,1)'
     1,'     dM(2,1)/dE'/)
      READ(4,3)
      READ(12,3)
      READ(14,3)
      READ(7,3)
      READ(3,3)
      READ(11,3)
3     FORMAT(/)
2     READ(12,1,END=1000)ITANK,(TMT(I),I=1,4)
      READ(14,1)ITANK,(FM(I),I=1,4)
      READ(3,1)ITANK,(FMM(I),I=1,4)
      READ(11,1)ITANK,(FMP(I),I=1,4)
1     FORMAT(2X,I2,2X,4(2X,D12.6))
```

```
      READ(4,1)ITANK,(TMTM(I),I=1,4)
      READ(7,1)ITANK,(TMTP(I),I=1,4)
      READ(15,7)ITANK,FREQ,ESYNC,EPS
7     FORMAT(2X,I2,3(1X,D12.6))
      SLPP=TMTP(2)/TMTP(1)
      SLPM=TMTM(2)/TMTM(1)
      SLP=TMT(2)/TMT(1)
C
C*** ASSUMES (DELTA E) = (1.005*ESYNC-0.995*ESYNC) = (ESYNC*.01)
C
      DSLP=(SLPP-SLPM)/(ESYNC*0.01)
      DM=(FMP(2)-FMM(2))/(ESYNC*0.01)
      WRITE(5,1)ITANK,SLP,DSLP,FM(2),DM
      WRITE(8,1)ITANK,SLP,DSLP,FM(2),DM
      GO TO 2
1000  STOP
      END
$
```

```
      SUBROUTINE EFSETS(ITANK,DTBI,FMSLP,DWA,DE)
C
C*********************************************************************
C
C  SUBROUTINE TO CALCULATE THE ELECTRIC FIELD DISPLACEMENT FROM
C  DESIGN BASED UPON A MEASUREMENT OF THE SLOPE OF THE LINE
C  OF ENERGY CHANGE VERSUS PHASE NEAR THE INTERSECTION OF
C  CURVE CLUSTERS IN DELTA-T PLANE(DESIGN PHASE IS USUALLY WITHIN A FEW DEGREES
C  OF THIS INTERSECTION).  AN ESTIMATE OF THE INPUT ENERGY
C  DISPLACEMENT IS ALSO GIVEN BASED UPON THE DELTA-TB VALUE AT
C  THE POINT OF INTERSECTION.
C
C  INPUTS:
C       ITANK   TANK NUMBER (11-17)
C       DTBI    MEASURED DELTA-T OF INTERSECTION (SECONDS)
C       FMSLP   (ENERGY CHANGE)/(PHASE CHANGE) MEASURED NEAR OR AT
C               THE DESIGN PHASE (OR NEAR THE INTERSECTION POINT IN
C               DELTA-T PLANE).
C
C  OUTPUTS:
C       DWA     ENERGY DISPLACEMENT (FRACTION) INTO TANK.
C       DE      FIELD DISPLACEMENT ESTIMATE (FRACTION).
C
C  INPUT FILES:
C       SLOPERS           -     RANDOM ACCESS FILE CONTAINING:
C               -ITANK  TANK NO.
C               -SLP    SLOPE OF DESIGN PARTICLE.
C               -DSLP   DERIVATIVE OF SLOPE WRT E FIELD.
C               -FM     2,1 ELEMENT OF TRANSFER MATRIX
C               -DFM    DERIVATIVE OF 2,1 ELEMENT OF TRANSFER
C                       MATRIX.
C
C       GTANKRS           -     RANDOM ACCESS FILE CONTAINING:
C               -ITNK   TANK NUMBER.
C               -EF     E FIELD (MV/M).
C               -DAB    TANK LENGTH (METERS).
C               -BTA    TANK INPUT BETA.
C
C            ***  WRITTEN BY T. L. OWENS  ***
C                     MARCH 25, 1991
C
C*********************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      OPEN(UNIT=4,FILE='SLOPERS.DAT',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      OPEN(UNIT=8,FILE='GTANKRS.DAT',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      READ(4,1,REC=ITANK-10)ITNK,SLP,DSLP,FM,DFM
1     FORMAT(2X,I2,2X,4(2X,D12.6))
      READ(8,2,REC=ITANK-10)ITNK,EF,DAB,BTA
2     FORMAT(2X,I2,3(2X,D12.6))
      WRITE(5,1)ITNK,SLP,DSLP,FM,DFM
      WRITE(5,2)ITNK,EF,DAB,BTA
      C=2.99792458D+8
      DE=(FMSLP-FM)/DFM
      GMA=DSQRT(1.0/(1.0-BTA**2))
      DBB=-DTBI*BTA*C/DAB
```

```
      DWA=DBB*GMA*(GMA+1.0)
      RETURN
      END
```

```
      SUBROUTINE PHSETS(ITANK,DTB,DTC,DPA,DWA)
C
C*****************************************************************
C
C  SUBROUTINE TO  CALCULATE THE PHASE AND ENERGY DISPLACEMENTS
C  FROM DESIGN VALUES.
C
C  INPUTS:
C       ITANK   TANK NUMBER (11-17).
C       DTB     DELTA-TB VALUE IN SECONDS.
C       DTC     DELTA-TC VALUE IN SECONDS.
C
C  OUTPUTS:
C       DPA     PHASE DISPLACEMENT (DEGREES).
C       DWA     ENERGY DISPLACEMENT (FRACTION).
C
C  INPUT FILES:
C       GTANKR.DAT      -       RANDOM ACCESS FILE CONTAINING:
C        -ITNK          TANK NUMBER.
C        -EF            E FIELD (MV/M).
C        -DAB           TANK LENGTH (M).
C        -BTA           INPUT BETA.
C
C       AMATR.DAT       -       RANDOM ACCESS FILE CONTAINING:
C        -ITNK          TANK NUMBER.
C        -A11-A22       ELEMENTS OF THE A MATRIX.
C
C
C
C                ***   WRITTEN BY T. L. OWENS   ***
C                           JAN 9, 1991
C
C*****************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      OPEN(UNIT=4,FILE='AMATRS.DAT',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      OPEN(UNIT=8,FILE='GTANKRS.DAT',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      READ(4,1,REC=ITANK-10)ITNK,A11,A21,A12,A22
1     FORMAT(2X,I2,2X,4(2X,D12.6))
      READ(8,2,REC=ITANK-10)ITNK,EF,DAB,BTA
2     FORMAT(2X,I2,3(2X,D12.6))
      WRITE(5,1)ITNK,A11,A21,A12,A22
      WRITE(5,2)ITNK,EF,DAB,BTA
      PI=4.0*DATAN(1.0D0)
      C=2.99792458D8
      EREST=939.301
      DPA=(A11*DTB+A12*DTC)*180.0/PI  !PHASE DISPLACEMENT IN DEGREES
      GMA=DSQRT(1.0/(1.0-BTA**2))
      WA=EREST*(GMA-1.0)
      DWA=(A21*DTB+A22*DTC)/WA        !ENERGY DISPLACEMENT FRACTION
      WRITE(5,3)WA
3     FORMAT(' WA=',D12.6)
      RETURN
      END
```

```
      SUBROUTINE EFSET2(ITANK,WP1,WP2,ECHNG,DE)
C
C*********************************************************************
C
C  SUBROUTINE TO CALCULATE THE ELECTRIC FIELD DISPLACEMENT FROM
C  DESIGN VALUES FOR TANKS 5-9, USING DELTA-T METHOD # 2.
C
C  INPUTS:
C       ITANK   TANK NUMBER.
C       WP1     INITIAL PEAK IN ENERGY VRS PHASE CURVE.
C       WP2     ENERGY PEAK FOR SLIGHTLY DIFFERENT E FIELD
C       ECHNG   CHANGE IN E FIELD (E2-E1)/E1 (FRACTION).
C
C  OUTPUTS:
C       DE      ELECTRIC FIELD DISPLACEMENT FROM DESIGN (FRACTION).
C
C  INPUT FILES:
C       SLOPER.DAT      -       RANDOM ACCESS FILE CONTAINING:
C        -ITNK          TANK NUMBER.
C        -FM21          2,1 ELEMENT OF THE TRANSFER MATRIX.
C
C       WPEAKR.LIS      -       RANDOM ACCESS FILE CONTAINING:
C        -ITNK          TANK NUMBER.
C        -WPS           PEAK ENERGY CHANGE FOR SYNCHRONOUS PARTICLE.
C        -DPHS          PHASE DISPLACEMENT OF PEAK ENERGY.
C
C                    ***   WRITTEN BY T. L. OWENS   ***
C                              JAN. 9, 1991
C
C*********************************************************************
C
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      OPEN(UNIT=4,FILE='SLOPER.DAT',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      OPEN(UNIT=9,FILE='WPEAKR.LIS',ACCESS='DIRECT',FORM='FORMATTED'
     1,STATUS='OLD',RECL=62)
      READ(4,1,REC=ITANK-2)ITNK,SLP,DSLP,FM21,DFM21
1     FORMAT(2X,I2,4(2X,D12.6))
      READ(9,3,REC=ITANK-4)ITNK,WPS,DPHS
3     FORMAT(2X,I2,2(2X,D12.6))
      WRITE(5,1)ITNK,SLP,DSLP,FM21,DFM21
      WRITE(5,3)ITNK,WPS,DPHS
C
C*** DWP IS THE TOTAL DERIVATIVE OF PEAK ENERGY WRT E.
C*** 0.624869352 IS TAN(-32 DEGREES).
C
      DWP=(WP2-WP1)/ECHNG+FM21/0.624869352
      DE=(WP1-WPS)/DWP
      RETURN
      END
```